

# Calculating ATR in Excel: EMA & Wilder's Methods

This is the second part of the Average True Range (ATR) Excel tutorial. In the [first part](#) we have calculated ATR using the simple moving average method. Now we will calculate ATR using two other popular [methods](#) – exponential moving average and Wilder's smoothing method.

**On this page:**[First Part Recap \(SMA ATR\)](#)[Exponential Moving Average ATR](#)[Excel Implementation of EMA ATR](#)[IF Conditions](#)[ATR Formulas](#)[Wilder's Smoothing Method ATR](#)[The Best Method and Period](#)

## First Part Recap (SMA ATR)

From the first part we have a spreadsheet with [historical data](#) in columns A-E, [true range](#) in column F and [SMA ATR](#) in column G.

## macroption

2							5
3	Date	Open	High	Low	Close	TR	ATR SMA
4	2017-06-01	120.17	120.74	120.06	120.72		
5	2017-06-02	121.31	121.71	121.19	121.61	0.99	
6	2017-06-05	121.91	121.94	121.58	121.73	0.36	
7	2017-06-06	123.09	123.31	122.77	123.1	1.58	
8	2017-06-07	122.79	123.16	122.02	122.45	1.14	
9	2017-06-08	122.07	122.08	120.95	121.65	1.5	1.114
10	2017-06-09	120.46	120.91	120.4	120.54	1.25	1.166
11	2017-06-12	120.33	120.69	120.27	120.36	0.42	1.178
12	2017-06-13	119.95	120.71	119.92	120.48	0.79	1.02
13	2017-06-14	121.51	121.88	119.57	119.82	2.31	1.254

Now we will calculate *exponential* moving average ATR in column H.

## Exponential Moving Average ATR

This method calculates ATR as exponential moving average of true range, in the same way exponential moving average of closing price works.

Each bar's ATR is calculated as weighted average of two inputs:

- The current bar's true range
- The previous bar's ATR

The formula is:

$$ATR = a \cdot TR + (1 - a) \cdot ATR_1$$

... where  $TR$  is current bar's true range,  $ATR_1$  is previous bar's ATR and  $a$  is the smoothing factor, which is calculated from the [ATR period input](#) ( $n$ ):

$$a = 2 / (n + 1)$$

For example, when you set ATR period to 14, the smoothing factor will be:

$$a = 2 / (14 + 1) = 2 / 15 = 0.1333$$

... and each bar's ATR will be:

## macroption

In other words, ATR will be composed of roughly 13% of current bar's true range and 87% of previous bar's ATR. These weights will change with ATR period – the shorter the period, the greater the weight of current bar's true range and the faster ATR reacts to changing market conditions.

There is one problem though. When each bar's ATR is calculated using the previous bar's ATR, how do we calculate ATR for the first bar? The answer is arithmetic average – exactly the same we did in the SMA method calculation (this is why the tutorial does the SMA method first).

## Excel Implementation of EMA ATR

Let's place ATR period input in cell H2 in our spreadsheet.

	A	B	C	D	E	F	G	H
1								
2							5	5
3	Date	Open	High	Low	Close	TR	ATR SMA	ATR EMA
4	2017-06-01	120.17	120.74	120.06	120.72			
5	2017-06-02	121.31	121.71	121.19	121.61	0.99		

The formulas in column H will use IF Excel functions and perform three different calculations, based on which row it is:

1. In the first  $n$  rows (where  $n$  is the ATR period) there is not enough data for ATR calculation, so the formula should return "" – empty cell (alternatively you can set it to return #N/A or something else).
2. In the  $n+1$ 'th row calculate arithmetic average of true range from all rows up to this one.
3. In all the subsequent rows, calculate ATR from previous row's ATR and current row's true range, using the formula explained above.

## IF Conditions

Let's first create the IFs and get the conditions right. The formula in cell H5 is:

```
=IF (ROW (F5) < 4 + H$2, "", IF (ROW (F5) = 4 + H$2, "SMA", "EMA"))
```

Instead of actual calculations, I'm using the placeholders "SMA" and "EMA" for now.

## macroption

...period, the number 4 corresponds to the position of our data. For example, if the ATR period is set to 5, this condition becomes current row < 4 + 5 and the formula will return "" in all rows above row 9. Starting from row 9, the condition will no longer be true and the next part of the formula will be executed.

The next part is another IF function. This time the condition checks whether the current row is equal to 4 + ATR period. If it is, we will calculate simple moving average (arithmetic average) of true range. If it isn't (therefore it must be greater), we will calculate the EMA ATR formula.

With the placeholders "SMA" and "EMA" and ATR period set to 5, the results look like this:

	A	B	C	D	E	F	G	H
1								
2							5	5
3	Date	Open	High	Low	Close	TR	ATR SMA	ATR EMA
4	2017-06-01	120.17	120.74	120.06	120.72			
5	2017-06-02	121.31	121.71	121.19	121.61	0.99		
6	2017-06-05	121.91	121.94	121.58	121.73	0.36		
7	2017-06-06	123.09	123.31	122.77	123.1	1.58		
8	2017-06-07	122.79	123.16	122.02	122.45	1.14		
9	2017-06-08	122.07	122.08	120.95	121.65	1.5	1.114	SMA
10	2017-06-09	120.46	120.91	120.4	120.54	1.25	1.166	EMA
11	2017-06-12	120.33	120.69	120.27	120.36	0.42	1.178	EMA
12	2017-06-13	119.95	120.71	119.92	120.48	0.79	1.02	EMA
13	2017-06-14	121.51	121.88	119.57	119.82	2.31	1.254	EMA

## ATR Formulas

Now we only need to replace "SMA" and "EMA" with the formulas to get the correct results.

The SMA formula is very simple – we can use the formula from the SMA ATR method in the [first part of this tutorial](#). In the formula in cell H5, we will replace "SMA" with this:

```
AVERAGE ( OFFSET ( F5, 1-H$2, 0, H$2, 1 ) )
```

Note: Unlike the SMA ATR method, in this case we know the range will always start in row 5, so we could actually use a simpler formula:

```
AVERAGE ( H$5 : H5 )
```

## macroption

If you have chosen the first version, the entire formula in cell H5 becomes:

```
=IF (ROW (F5) < 4 + H$2, "", IF (ROW (F5) = 4 + H$2, AVERAGE (OFFSET (F5, 1 - H$2, 0, H$2, 1)), "EMA"))
```

You can copy this formula to the other rows to check the results. If you have the IF conditions right and ATR period set to 5, you should see empty cells in rows 5-8, then the number 1.114 (the 5-period SMA ATR) in cell H9, and "EMA" in the rows below.

	A	B	C	D	E	F	G	H
1								
2							5	5
3	Date	Open	High	Low	Close	TR	ATR SMA	ATR EMA
4	2017-06-01	120.17	120.74	120.06	120.72			
5	2017-06-02	121.31	121.71	121.19	121.61	0.99		
6	2017-06-05	121.91	121.94	121.58	121.73	0.36		
7	2017-06-06	123.09	123.31	122.77	123.1	1.58		
8	2017-06-07	122.79	123.16	122.02	122.45	1.14		
9	2017-06-08	122.07	122.08	120.95	121.65	1.5	1.114	1.114
10	2017-06-09	120.46	120.91	120.4	120.54	1.25	1.166	EMA
11	2017-06-12	120.33	120.69	120.27	120.36	0.42	1.178	EMA
12	2017-06-13	119.95	120.71	119.92	120.48	0.79	1.02	EMA
13	2017-06-14	121.51	121.88	119.57	119.82	2.31	1.254	EMA

Let's now replace "EMA" with a new EMA ATR formula:

$$ATR = a \cdot TR + (1 - a) \cdot ATR_1$$

$$ATR = (2 / (n + 1)) \cdot TR + (1 - 2 / (n + 1)) \cdot ATR_1$$

In the formula in cell H5, we will replace "EMA" with:

```
2 / (H$2 + 1) * F5 + (1 - 2 / (H$2 + 1)) * H4)
```

The entire formula in cell H5 becomes:

```
=IF (ROW (F5) < 4 + H$2, "", IF (ROW (F5) = 4 + H$2, AVERAGE (OFFSET (F5, 1 - H$2, 0, H$2, 1)), 2 / (H$2 + 1) * F5 + (1 - 2 / (H$2 + 1)) * H4))
```

## macroption

If you copy the formula from H5 to the other rows, you should get these results for ATR period 5 and the GLD daily data we are using:

	A	B	C	D	E	F	G	H
1								
2							5	5
3	Date	Open	High	Low	Close	TR	ATR SMA	ATR EMA
5	2017-06-02	121.31	121.71	121.19	121.61	0.99		
6	2017-06-05	121.91	121.94	121.58	121.73	0.36		
7	2017-06-06	123.09	123.31	122.77	123.1	1.58		
8	2017-06-07	122.79	123.16	122.02	122.45	1.14		
9	2017-06-08	122.07	122.08	120.95	121.65	1.5	1.114	1.114
10	2017-06-09	120.46	120.91	120.4	120.54	1.25	1.166	1.159333
11	2017-06-12	120.33	120.69	120.27	120.36	0.42	1.178	0.912889
12	2017-06-13	119.95	120.71	119.92	120.48	0.79	1.02	0.871926
13	2017-06-14	121.51	121.88	119.57	119.82	2.31	1.254	1.351284
14	2017-06-15	119.22	119.5	119.13	119.32	0.69	1.092	1.130856
15	2017-06-16	119.35	119.5	119.24	119.34	0.26	0.894	0.840571
16	2017-06-19	118.74	118.84	118.33	118.43	1.01	1.012	0.897047
17	2017-06-20	118.42	118.46	118.07	118.18	0.39	0.932	0.728031

## Wilder's Smoothing Method ATR

The last of the three methods that remains is Wilder's smoothing method. This is the one presented in [New Concepts in Technical Trading Systems](#), the book by J. Welles Wilder that first introduced ATR to the world.

It is in fact almost the same as the EMA ATR method explained above, with just one difference.

Under the EMA method the smoothing factor  $a$  is calculated as  $2 / (n + 1)$ . Under Wilder's method it is simply  $1 / n$ . The ATR formula becomes:

$$ATR = (1/n) \cdot TR + (1 - 1/n) \cdot ATR_1$$

Therefore, we can use the formula from the EMA method and only change the smoothing factor to get Wilder's ATR. The formula in cell I5 is:

```
=IF(ROW(F5)<4+I$2,"",IF(ROW(F5)=4+I$2,AVERAGE(OFFSET(F5,1-I$2,0,I$2,1)),1/I$2*F5+(1-1/I$2)*I4))
```

## macroption

$$1/I\$2 * F5 + (1 - 1/I\$2) * I4$$

The results with our data look like this:

	A	B	C	D	E	F	G	H	I
1									
2							5	5	5
3	Date	Open	High	Low	Close	TR	ATR SMA	ATR EMA	ATR Wilder
4	2017-06-01	120.17	120.74	120.06	120.72				
5	2017-06-02	121.31	121.71	121.19	121.61	0.99			
6	2017-06-05	121.91	121.94	121.58	121.73	0.36			
7	2017-06-06	123.09	123.31	122.77	123.1	1.58			
8	2017-06-07	122.79	123.16	122.02	122.45	1.14			
9	2017-06-08	122.07	122.08	120.95	121.65	1.5	1.114	1.114	1.114
10	2017-06-09	120.46	120.91	120.4	120.54	1.25	1.166	1.159333	1.1412
11	2017-06-12	120.33	120.69	120.27	120.36	0.42	1.178	0.912889	0.99696
12	2017-06-13	119.95	120.71	119.92	120.48	0.79	1.02	0.871926	0.955568
13	2017-06-14	121.51	121.88	119.57	119.82	2.31	1.254	1.351284	1.226454
14	2017-06-15	119.22	119.5	119.13	119.32	0.69	1.092	1.130856	1.119164
15	2017-06-16	119.35	119.5	119.24	119.34	0.26	0.894	0.840571	0.947331
16	2017-06-19	118.74	118.84	118.33	118.43	1.01	1.012	0.897047	0.959865
17	2017-06-20	118.42	118.46	118.07	118.18	0.39	0.932	0.728031	0.845892

## The Best Method and Period

You can see that the three methods give quite different results even with the same period setting. This is not a surprise when you look at the formulas.

You may be wondering which method and which period are the "correct" ones.

Regarding ATR period, popular ones appear to be 14 and 20, but there is no magic number. Consistency and suitability to your particular purpose and conditions are more important than the exact value you choose. Furthermore, discussing the best period doesn't make any sense without also specifying which method we are using, because the period's effect is different across methods. [More here](#).

Regarding the calculation methods, obviously the "correct" method is the one presented by the original author of ATR, J. Welles Wilder. That said, my experience from using various financial or charting software is that all the three methods are commonly used. One problem is that many

## macroption

yourself and compare other party results with your own. This is also the reason that first led me to developing the [ATR Calculator](#).

---

[All](#) » [Tutorials and Reference](#) » [Technical Analysis](#) » [Average True Range \(ATR\)](#)

[What Is ATR and Why Use It in Trading](#)

[True Range and How It Differs from Range](#)

[ATR Calculation Methods and Formulas](#)

[Calculating Average True Range \(ATR\) in Excel](#)

[Calculating ATR in Excel: EMA & Wilder's Methods](#)

[Normalized ATR: Two Ways of Expressing ATR as Percentage](#)

[Average True Range \(ATR\) Period and Which to Use](#)

**More in Technical Analysis**

[Average True Range \(ATR\)](#)

[Relative Strength Index \(RSI\)](#)

[Moving Averages](#)

[MACD \(Moving Average Convergence-Divergence\)](#)

**Excel Calculators**

[Option Strategy Payoff Calculator](#)

[Option Strategy Simulator](#)

[Option Portfolio Manager](#)

[Price Probability Calculator](#)

[Black-Scholes Calculator](#)

[Binomial Option Pricing Calculator](#)

[Historical Volatility Calculator](#)

[Implied Volatility Calculator](#)

[Average True Range \(ATR\) Calculator](#)

[Relative Strength Index \(RSI\) Calculator](#)

[Futures Price Calculator](#)

[Descriptive Statistics Calculator](#)

[Top of this page](#)

[Home](#)

[Tutorials](#)

[Calculators](#)

[Services](#)

[About](#)

[Contact](#)

By remaining on this website or using its content, you confirm that you have read and agree with the [Terms of Use Agreement](#).



# macroption

Content may include affiliate links, which means we may earn commission if you buy on the linked website. See full [Affiliate and Referral Disclosure](#).

We use cookies and similar technology to improve user experience and analyze traffic. See full [Cookie Policy](#).

See also [Privacy Policy](#) on how we collect and handle user data.

© 2023 Macroption